
By the end of today you will read every line

Billingsley §3 (uniqueness of measures)
Two probability measures that agree on a π-system generating the σ-field agree everywhere.

The proof, checked by a computer, line by line:

-- To show μ = ν it suffices to check μ t = ν t for every measurable t.
ext t ht

-- "Let L be the class of sets where μ and ν agree; L is a λ-system ⊇ P."

induction t, ht using MeasurableSpace.induction_on_inter hgen hP with

| empty => simp

| basic t ht => exact hagree t ht

| compl t htm ih =>

-- P(Aᶜ) = 1 − P(A): no subtraction on ℝ≥0∞ without this lemma

rw [prob_compl_eq_one_sub htm, prob_compl_eq_one_sub htm, ih]

| iUnion f hdisj hfm ih =>

-- countable additivity on a disjoint union

rw [measure_iUnion hdisj hfm, measure_iUnion hdisj hfm]

simp [ih]

I’m learning with you

Le (and you)
a probabilist, not a Lean expert:
writes and reads the statement

AI assistant
drafts proofs, suggests lem-
mas, explains error messages

Lean + Mathlib
the referee:

accepts or rejects every step

statement
+ question

proof attempt

X or an
exact error

House rules for this class
I Every Lean code block on these slides is cut from a file that compiles against the course pin (Lean v4.34.0,

Mathlib v4.34.0).
I When I don’t know, I say so, and we ask Lean together.
I My own Lean self-grade today: 50/100. Let’s raise it together.
I Through-line: Lean checks every step; you check the statement.

Lean in 2026: how we got here

future

2017 2019 2021 2022 2023 2024 2025 2026 2027

mathlib
community library

starts (Lean 3)

Liquid Tensor
Scholze’s chal-
lenge theorem

checked (Jul 2022)

PFR: theorem
of Gowers–

Green–Manners–Tao
(Nov 2023),
formalized

in ~3 weeks

Fermat’s Last
Theorem

multi-year project led
by Buzzard (2024–)

AI provers
Lean proofs of

olympiad problems
(2022–); IMO silver

level (Jul 2024)

Navier–Stokes blow-
up (forced) (Ope-

nAI, Sep 2026)
~10k agents, ~88h

search + ~17h Lean
machine-checked, not

yet peer-reviewed

you, today (Sep 2026)

Mathlib today: most of the measure theory of Billingsley Chapters 2–3 is already there. Today we learn to read it.

A theorem has three parts

theorem uniqueness_of_extension {Ω : Type*} [m : MeasurableSpace Ω]
(μ ν : Measure Ω) [IsProbabilityMeasure μ] [IsProbabilityMeasure ν]

(P : Set (Set Ω)) (hgen : m = MeasurableSpace.generateFrom P)

(hP : IsPiSystem P) (hagree : ∀ s ∈ P, μ s = ν s) : μ = ν := by

Statement
everything between
theorem and :=

hypotheses, then : conclusion

Proof
everything after :=

(here: by + a list of tactics)

Kernel
a small program that checks

the proof proves this statement

checked by you checked by Lean

I (hP : IsPiSystem P): a hypothesis with a name you can use later. [IsProbabilityMeasure μ]: an
instance, found automatically. {Ω : Type*}: inferred.

The Infoview: what you have, what you must show

Lean Infoview

H Tactic state 1 goal

Ω : Type u_1

m : MeasurableSpace Ω

μ ν : Measure Ω

inst✝¹ : IsProbabilityMeasure μ

inst✝ : IsProbabilityMeasure ν

P : Set (Set Ω)

hgen : m = MeasurableSpace.generateFrom P

hP : IsPiSystem P

co
nt

ex
t

hagree : ∀ s ∈ P, μ s = ν s

t : Set Ω

ht : MeasurableSet t

⊢ μ t = ν t

above ⊢: what you have
one hypothesis per
line, name : type

after ⊢: what you must show

inst✝: a dagger means “exists,
but you cannot type its name”

Mock-up; the text is the real state after ext t ht in Uniqueness.lean.

sorry: what “checked” means

Red: error
the statement or a step

does not make sense
e.g. a conclusion of the

wrong type (live in Demo 0)

Yellow: sorry

statement type-checks;
the proof is a promise
warning: declaration

uses `sorry`

Green: checked
no messages at all; every

step accepted by the kernel
(no output)

fix the statement fill the proof

A sorry cannot hide
#print axioms on a theorem that depends on a sorry anywhere upstream prints an axiom list; here the list
contains sorryAx. Even a failed proof is turned into a hidden sorryAx, so the axiom list catches it. The warm-up
file starts with every proof equal to sorry; your job is to replace them.

The trust chain

Kernel
a small pro-

gram, checked
by many people

Axioms
propext,

Classical.choice,
Quot.sound

Proof
every tactic step,
every lemma used

Statement
and the defini-

tions it mentions

X Lean X Lean X Lean ? you

If the statement says something weaker than the book, Lean will happily certify the weaker thing.
Lean checks every step; you check the statement.

Live example (Sep 2026)
The Lean-checked Navier–Stokes proof is of forced finite-time blow-up (Clay (C)/(D), smooth external force); the
unforced problem is open and the proof is not yet peer-reviewed. Lean checked the proof; people check the
statement.

Install in five steps (announcement of 9/23)

1. elan
Lean version

manager
Win/macOS: the

extension installs it

2. VS Code
+ the Lean
4 extension
(leanprover)

3. Unzip
to a plain path

C:\lean\Math7800-

Lean

~/lean/Math7800-

Lean

4. Cache
open the folder;
in its terminal:

lake exe cache get

5. Open
Uniqueness.lean

wait until the
orange bar
disappears

downloads prebuilt Mathlib
(about 450 MB, about 8 GB
on disk); do it on good Wi-Fi

no OneDrive, no
iCloud, no spaces

The one rule people forget
Open the folder (the one with lakefile.toml), not a single .lean file. Otherwise Lean reports unknown package

Mathlib. Blank Infoview: Command Palette → “Lean 4: Restart Server”. More pitfalls per system: README.md

and the backup slides.

Verify your install, or use the browser for now

Verify (30 seconds)
1. Open Math7800Lean/Uniqueness.lean.
2. Click just after ext t ht.
3. The Infoview ends with ⊢ μ t = ν t.
4. Screenshot it: that is your proof the install works.

Browser fallback
1. Go to live.lean-lang.org.
2. Paste the file’s contents; wait.
3. Version picker: choose the Stable release preset if

offered.
For reading and experimenting only. Check anything you
submit on your own install or in office hours.

goal visible:
you are ready

not yet: browser today,
fix it by Thursday

office hours
Tue/Thu 1:00–1:50,

STEM+Ag 3147

or

A dictionary: Billingsley ↔ Mathlib

Billingsley Mathlib Note

sample space Ω Ω : Type* a type; subsets are Set Ω

class of subsets A Set (Set Ω)

σ-field F on Ω m : MeasurableSpace Ω usually an instance

A ∈ F MeasurableSet A MeasurableSet[m] A names it

measure µ, value µ(A) μ : Measure Ω, μ A values in ℝ≥0∞ = [0,∞]

probability measure P [IsProbabilityMeasure μ] measure_univ : μ univ = 1

Every Mathlib name here was checked with #check at the course pin. More rows (fields, σ(A), u, the lemmas of our proof): backup
glossary.

Warm-up 1: inclusion–exclusion, the Mathlib way

/-- Billingsley §2: P(A ∪ B) + P(A ∩ B) = P(A) + P(B). -/
theorem union_add_inter (μ : Measure Ω) (A B : Set Ω) (hB : MeasurableSet B) :

μ (A ∪ B) + μ (A ∩ B) = μ A + μ B :=

measure_union_add_inter A hB

Goal at the sorry in the student file:

Ω : Type u_1

inst✝ : MeasurableSpace Ω

μ : Measure Ω

A B : Set Ω

hB : MeasurableSet B

⊢ μ (A ∪ B) + μ (A ∩ B) = μ A + μ B

Two things to notice
I Stated with + on both sides, not

P(A ∪ B) = P(A) + P(B) − P(A ∩ B):
subtraction in ℝ≥0∞ is truncated (next frame).

I Only B must be measurable: a measurable set
splits every set additively.

Warm-ups 3–5 (measure_mono', prob_le_one', compl_prob): exact? finds each one. You will try one yourself in
the demo.

The ℝ≥0∞ trap: subtraction is truncated

0 1 2 ∞ = ⊤

1 − 2 = 0 ∞ − ∞ = 0

So µ(Ac) = µ(Ω) − µ(A) needs µ(A) 6= ∞. The general lemma measure_compl asks for it; the probability
lemma has it built in:

| compl t htm ih =>
-- P(Aᶜ) = 1 − P(A): no subtraction on ℝ≥0∞ without this lemma

rw [prob_compl_eq_one_sub htm, prob_compl_eq_one_sub htm, ih]

Forget it and Lean answers with two leftover goals
(μ t ≠ ⊤, ν t ≠ ⊤), not “wrong”. You’ll see this live in
Demo 3. Rule of thumb

For probabilities use prob_compl_eq_one_sub; write
inclusion–exclusion additively.

How to find a lemma

What do you know?

the name

you are in a goal

the shape of
the statement

it in words

#check name on its
own line, or hover

exact?: searches Math-
lib, offers a line to click

Loogle: loogle.lean-lang.org

LeanSearch: leansearch.net

You’ll see exact? live in Demo 3, and use it yourself on a warm-up.

The definitions on paper

π-system (paraphrasing Billingsley §3)
A class P of subsets of Ω is a π-system if it is closed under finite intersections: A, B ∈ P =⇒ A ∩ B ∈ P.

λ-system (paraphrasing Billingsley §3)
A class L of subsets of Ω is a λ-system if
(λ1) Ω ∈ L;
(λ2) A ∈ L =⇒ Ac ∈ L;
(λ3) A1, A2, . . . ∈ L pairwise disjoint =⇒

⋃
n An ∈ L.

Generated classes (paraphrasing Billingsley §3)
σ(P) and λ(P): the smallest σ-field and the smallest λ-system containing P (the intersection of all of them).

The same definition, twice in Lean

Our midterm (Billingsley’s version)

/-- λ-system (Billingsley): contains Ω, closed under

complements and↪→
countable disjoint unions. -/

structure IsLambdaSys (L : Set (Set Ω)) : Prop where

univ_mem : univ ∈ L

compl_mem : ∀ A ∈ L, Aᶜ ∈ L

iUnion_mem : ∀ f : ℕ → Set Ω, Pairwise (Disjoint on f) →

(∀ n, f n ∈ L) →↪→
(� n, f n) ∈ L

Mathlib: MeasurableSpace.DynkinSystem

has_empty: contains ∅
has_compl: closed under complements
has_iUnion_nat: countable pairwise disjoint
unions

Billingsley: Ω ∈ L Mathlib: ∅ ∈ L
∅ = Ωc

Ω = ∅c

Also different: Mathlib’s IsPiSystem only requires closure when A ∩ B 6= ∅.

Checking the statement includes checking the defini-
tions: Ω vs ∅, the nonempty clause. This is audit point (1).

The picture: P ⊆ λ(P) = σ(P) ⊆ L

any λ-system L ⊇ P (e.g. {A measurable : µA = νA})

λ(P) = σ(P) when P is a π-system

P (π-system)

λ(P) ⊆ L: minimality

σ(P) ⊆ λ(P):
the work (“good sets”)

Every σ-field is a λ-system, so λ(P) ⊆ σ(P) is free; the content is the other inclusion.

The π-λ theorem, and how Mathlib packages it

π-λ theorem (paraphrasing Billingsley §3)
If P is a π-system and L is a λ-system with P ⊆ L, then σ(P) ⊆ L.

Uniqueness (paraphrasing Billingsley §3)
If probability measures µ, ν agree on a π-system P, they agree on σ(P).

Mathlib: MeasurableSpace.induction_on_inter, an induction
principle
If m = σ(s) with s a π-system, and a property C of measurable sets

holds for ∅ holds on s passes to Ac passes to dis-
joint

⋃
n

holds for every
measurable set

The four boxes are exactly “{A measurable : C(A)} is a (Dynkin) λ-system containing s”.

Two tactics do the architecture

-- To show μ = ν it suffices to check μ t = ν t for every measurable t.
ext t ht

-- "Let L be the class of sets where μ and ν agree; L is a λ-system ⊇ P."

induction t, ht using MeasurableSpace.induction_on_inter hgen hP with

⊢ μ = ν
t, ht

⊢ μ t = ν t

ext

empty

basic

compl

iUnion

induction … using induction_on_inter

µ∅ = ν∅ (book: Ω ∈ L)

P ⊆ L

L closed under c

L closed under disjoint
⋃

L = {A measurable : µA = νA}. The one induction line is the π-λ theorem; it leaves four goals.

Cases empty and basic: ∅ ∈ L, P ⊆ L

induction t, ht using MeasurableSpace.induction_on_inter hgen hP with
| empty => simp

| basic t ht => exact hagree t ht

| compl t htm ih =>

-- P(Aᶜ) = 1 − P(A): no subtraction on ℝ≥0∞ without this lemma

rw [prob_compl_eq_one_sub htm, prob_compl_eq_one_sub htm, ih]

| iUnion f hdisj hfm ih =>

-- countable additivity on a disjoint union

rw [measure_iUnion hdisj hfm, measure_iUnion hdisj hfm]

simp [ih]

Goals before simp, and before exact:

case empty

[CTX]

t : Set Ω

⊢ μ ∅ = ν ∅

case basic

[CTX]

t✝ t : Set Ω

ht : t ∈ P

⊢ μ t = ν t

[CTX] = the nine context lines of frame 2b, down to hagree.

Paper (paraphrasing Billingsley §3): µΩ = 1 = νΩ,
and P ⊆ L by hypothesis. Mathlib’s form checks ∅:
µ∅ = 0 = ν∅.

simp rewrites with library lemmas; here measure_empty : μ

∅ = 0 on both sides. hagree is a function: feed it a set t and
a proof ht : t ∈ P, get μ t = ν t. exact says “this term is
the proof”.

Case compl: L is closed under complements

induction t, ht using MeasurableSpace.induction_on_inter hgen hP with
| empty => simp

| basic t ht => exact hagree t ht

| compl t htm ih =>

-- P(Aᶜ) = 1 − P(A): no subtraction on ℝ≥0∞ without this lemma

rw [prob_compl_eq_one_sub htm, prob_compl_eq_one_sub htm, ih]

| iUnion f hdisj hfm ih =>

-- countable additivity on a disjoint union

rw [measure_iUnion hdisj hfm, measure_iUnion hdisj hfm]

simp [ih]

Goal before, then after each rewrite:

case compl

[CTX]

t✝ t : Set Ω

htm : MeasurableSet t

ih : μ t = ν t

⊢ μ tᶜ = ν tᶜ

⊢ 1 - μ t = ν tᶜ

⊢ 1 - μ t = 1 - ν t

Paper (paraphrasing Billingsley §3): if A ∈ L then
µAc = 1 − µA = 1 − νA = νAc .

rw [a, b, c] is three rewrites. The third (ih) makes both
sides 1 - ν t, and rw closes an identical goal by itself. ih:
µt = νt is the induction hypothesis: C already holds for t;
we show it for tc .

Case iUnion: countable disjoint unions

induction t, ht using MeasurableSpace.induction_on_inter hgen hP with
| empty => simp

| basic t ht => exact hagree t ht

| compl t htm ih =>

-- P(Aᶜ) = 1 − P(A): no subtraction on ℝ≥0∞ without this lemma

rw [prob_compl_eq_one_sub htm, prob_compl_eq_one_sub htm, ih]

| iUnion f hdisj hfm ih =>

-- countable additivity on a disjoint union

rw [measure_iUnion hdisj hfm, measure_iUnion hdisj hfm]

simp [ih]

Goal before, then after each rewrite:

case iUnion

[CTX] context

t : Set Ω

f : ℕ → Set Ω

hdisj : Pairwise (Function.onFun Disjoint f)

hfm : ∀ (i : ℕ), MeasurableSet (f i)

ih : ∀ (i : ℕ), μ (f i) = ν (f i)

⊢ μ (� i, f i) = ν (� i, f i)

⊢ ∑' (i : ℕ), μ (f i) = ν (� i, f i)

⊢ ∑' (i : ℕ), μ (f i) = ∑' (i : ℕ), ν (f i)

Paper (paraphrasing Billingsley §3): if An ∈ L
are disjoint, µ(

⋃
An) =

∑
µAn =

∑
νAn =

ν(
⋃

An).

measure_iUnion is countable additivity; hdisj, hfm
are its hypotheses. simp [ih] rewrites inside the sum,
under the bound i, where rw cannot.

Payoff: a law on R is determined by its CDF

/-- Payoff: a law on ℝ is determined by its CDF. -/
theorem law_eq_of_cdf_eq (μ ν : Measure ℝ) [IsProbabilityMeasure μ]

[IsProbabilityMeasure ν] (h : ∀ x, μ (Iic x) = ν (Iic x)) : μ = ν := by

refine uniqueness_of_extension μ ν (range Iic) ?_ isPiSystem_Iic ?_

· exact BorelSpace.measurable_eq.trans (borel_eq_generateFrom_Iic ℝ)

· rintro _ ⟨x, rfl⟩

exact h x

Plug P = half-lines (−∞, x ] into our theorem; two small facts remain.

What we just did

Case Paper (Billingsley §3) Lean

empty Ω ∈ L (Mathlib: ∅) simp (uses measure_empty)

basic P ⊆ L by hypothesis exact hagree t ht

compl µAc = 1 − µA = 1 − νA = νAc rw [prob_compl_eq_one_sub htm, …, ih]

iUnion µ
⋃

An =
∑

µAn =
∑

νAn rw [measure_iUnion …]; simp [ih]

paper proof:
one paragraph

Lean proof:
about ten lines, ev-

ery step checked

same argument

Questions? (pause here)

How I actually work

1. Ask
I write the state-

ment, then
ask for a proof

2. Proof
AI drafts; Lean

compiles it

3. Audit
the 3 points
(next slide)

4. Understand
read it line by

line, as we just did

pass

fail: statement changed, or a sorry

Lean error → paste it back

Why step 4 is not optional
Lean has already told you the proof is correct. Reading it is for you: the oral check asks you to explain each step,
and the next proof you write will reuse these moves.

The 3-point audit

(1) The statement is the one you meant. Compare it, character by character, with what
you asked for: hypotheses added or strengthened, conclusion weakened (≤ for =, ∃ for ∀),
a changed definition.

(2) No sorry. The file compiles with no declaration uses `sorry` warning (also look
for admit, new axiom).

(3) #print axioms is clean. Only propext, Classical.choice, Quot.sound (a subset is
fine); sorryAx or any other name fails.

you

Lean

Lean

Lean checks (2) and (3) for you; only you can check (1). The midterm checker runs all three, and point (1)
exactly: your theorem must type-check against the frozen statement.

25 / 29 (+ 6)



Exhibit: the AI changed a hypothesis

/-- The AI's version, compiled: the statement was changed. -/

theorem uniqueness_AI {Ω : Type*} [m : MeasurableSpace Ω]

(μ ν : Measure Ω) [IsProbabilityMeasure μ] [IsProbabilityMeasure ν]

(P : Set (Set Ω)) (hP : ∀ s, MeasurableSet s → s ∈ P)

(hagree : ∀ s ∈ P, μ s = ν s) : μ = ν := by

ext t ht

exact hagree t (hP t ht)

#print axioms uniqueness_AI

-- 'uniqueness_AI' depends on axioms:

-- [propext, Classical.choice, Quot.sound]

Real statement
hgen : m = generateFrom P

hP : IsPiSystem P

AI statement
hP : ∀ s, MeasurableSet s → s ∈ P

P already holds every measurable set

AI edit

(2) X (3) X (1) X (2) X (3) X (1) 7: trivial, useless for CDFs

Tools, and what Lean cannot tell you

Tools I use
I exact?, #check, hover: inside VS Code.
I Loogle, LeanSearch: find a Mathlib name.
I An AI chat: drafts, and explains error messages.
I Lean Zulip: real people, very helpful to

beginners.

Lean cannot tell you
I whether the statement matches the book;
I whether the definitions match the book;
I whether the result is interesting;
I whether you understand it.

Honesty: using AI is allowed in this course. Say how you used it. The write-
up is in your own words, and the oral check is you explaining your proof.

Take-home midterm: two Lean projects

Project A: the π-λ theorem
/-- Project A: Dynkin's π-λ theorem (Billingsley §3). -/

theorem pi_lambda (P L : Set (Set Ω)) (hP : IsPiSys P) (hL : IsLambdaSys L)

(hPL : P ⊆ L) : sigmaGen P ⊆ L := by

If P is a π-system, L a λ-system (Billingsley’s, with Ω) and P ⊆ L, then σ(P) ⊆ L. Scaffold: four lemmas
with sorry.

Project B: the monotone class theorem (not in Mathlib)
/-- Project B: monotone class theorem (Billingsley §3). -/

theorem monotone_class (F₀ : Set (Set Ω)) (hF₀ : IsField F₀) :

monoGen F₀ = sigmaGen F₀ := by

If F0 is a field, the monotone class it generates equals the σ-field it generates: m(F0) = σ(F0).

Statements are frozen: point (1) is enforced by the checker.

Part 1: Thu 10/1, no class — self-study (video + FPIL Ch. 1)

Part 2: A/B released today, end of class due Thu 10/29 Gradescope: two .lean files + one PDF

Rules, grading, and where to get help

Rules
I Learn with AI however you like (ChatGPT, Claude,

...): what matters is that you can read and write
Lean yourself.

I PDF write-up: every step in your own words, plus
an honest self-assessment (see below).

I 10-minute oral check on your proofs.
I Project A may not use anything in

MeasurableSpace.DynkinSystem.* or
MeasurableSpace.generateFrom*, nor
MeasurableSpace.induction_on_inter.

Help
I Office hours Tue/Thu 1:00–1:50, STEM+Ag

3147.
I Lean Zulip:

leanprover.zulipchat.com

I CHEATSHEET.md, Mathlib docs, Loogle.

Checker
compiles; statement

matches; axioms clean;
no forbidden lemma

Write-up
your own words

Oral
10 minutes

default 50%
default 25% default 25%

Self-assessment (end of write-up, our 100-pt scale): if you were the professor, how satisfied would you be with
yourself as a student? Le’s own grade today: 50/100.

Backup: the paper proof of the π-λ theorem

Let P be a π-system and L ⊇ P a λ-system (Billingsley’s argument, paraphrased).
1. λ(P) is a λ-system containing P, and λ(P) ⊆ L by minimality.
2. A λ-system contains ∅ = Ωc and is closed under finite disjoint unions (pad with ∅).
3. Good sets. For A ∈ λ(P) let GA = {B : A ∩ B ∈ λ(P)}. Then GA is a λ-system: Ω ∈ GA; if B ∈ GA

then A ∩ Bc =
(
Ac ∪ (A ∩ B)

)c , a complement of a disjoint union; disjoint unions pass through A ∩ ·.
4. Round 1. For A ∈ P: P ⊆ GA (π-system), so λ(P) ⊆ GA. Round 2. For A ∈ λ(P): round 1 gives

P ⊆ GA, so again λ(P) ⊆ GA. Hence λ(P) is closed under ∩.
5. A λ-system closed under ∩ is a σ-field: write

⋃
An =

⋃
Bn with Bn = An ∩ Ac

1 ∩ · · · ∩ Ac
n−1 disjoint.

6. So λ(P) is a σ-field containing P: σ(P) ⊆ λ(P) ⊆ L.

Project A’s four scaffold lemmas cover steps 1 (two lemmas), 3–4 and 5; step 6 is the main theorem.

Backup: tactic cheat sheet

Tactic What it does
intro introduce a variable or hypothesis for a ∀ / → goal
exact close the goal with a term that matches it
apply match the lemma’s conclusion; its hypotheses become goals
refine like exact, with ?_ holes left as goals
rw [h] rewrite the goal left-to-right with an equation
simp simplify with the library’s @[simp] lemmas
constructor split ∧, Iff, or a structure into its fields
obtain, rintro take apart ∃, ∧, ⟨_, _⟩

ext equality of measures/sets/functions, pointwise
induction … using induction with a custom principle
calc chain equalities and inequalities
exact? search Mathlib for a closing lemma
sorry placeholder; the file compiles with a warning

Full version with examples: student/CHEATSHEET.md.

Backup: error catalogue (all systems)

Symptom Fix
lake: command not found restart VS Code or the terminal after installing elan
Mathlib builds for hours you ran lake build first; Ctrl+C, then lake exe cache get

Windows: “filename too long” git config --global core.longpaths true

project in OneDrive / iCloud / path
with spaces

move the folder to C:\lean\Math7800-Lean or
~/lean/Math7800-Lean

unknown package Mathlib open the folder with lakefile.toml, not a single file
Infoview blank Command Palette: “Lean 4: Restart Server”
Infoview stuck on one file Command Palette: “Lean 4: Restart File”
macOS: tools missing xcode-select --install

Backup: more of the glossary

Billingsley Mathlib Note
field (algebra) of sets (course) IsField midterm only; not in Mathlib
σ(A) MeasurableSpace.generateFrom

F1 ∩ F2 m₁ � m₂ “inf” in a lattice
F1 ⊆ F2 m₁ ≤ m₂

π-system IsPiSystem only when A ∩ B 6= ∅
λ-system MeasurableSpace.DynkinSystem contains ∅, not Ω

π-λ theorem induction_on_inter an induction principle
finite measure [IsFiniteMeasure μ] automatic for probability
µ(Ac) = µ(Ω) − µ(A) measure_compl needs μ A ≠ ⊤

P(Ac) = 1 − P(A) prob_compl_eq_one_sub needs MeasurableSet A

µ(A) < ∞, finite µ measure_ne_top

countable additivity measure_iUnion sum is ∑'

A ⊆ B ⇒ µA ≤ µB measure_mono no measurability
countable subadditivity measure_iUnion_le

µ = ν iff equal on F Measure.ext applied by ext

(iff form) Measure.ext_iff

Borel sets from (−∞, x] borel_eq_generateFrom_Iic

distribution function F ProbabilityTheory.cdf μ

image measure µ ◦ f −1 Measure.map f μ

Backup: top two pitfalls on each system

Windows
1. Project in OneDrive, Desk-
top, or a path with spaces.
Fix: move it to
C:\lean\Math7800-Lean.

2. “filename too long”.
Fix: git config --global

core.longpaths true

macOS
1. Command-line tools missing.
Fix: xcode-select --install

first.

2. Project inside iCloud Drive
(synced folder).
Fix: use ~/lean/Math7800-

Lean.

Linux
1. lake: command not found.
Fix: open a new terminal (or
restart VS Code) after elan.

2. Ran lake build first: Math-
lib builds for hours.
Fix: Ctrl+C, then lake exe

cache get.

Blank Infoview on any system: Command Palette → “Lean 4: Restart Server”. Full table: README.md, and the
backup slide “Error catalogue”.

Backup: Warm-up 2, a pair of proofs, 〈h1, h2〉

/-- The intersection of two σ-fields is a σ-field (Lean: `m₁ � m₂`). -/
theorem inf_measurableSet (m₁ m₂ : MeasurableSpace Ω) (s : Set Ω)

(h₁ : MeasurableSet[m₁] s) (h₂ : MeasurableSet[m₂] s) :

MeasurableSet[m₁ � m₂] s :=

⟨h₁, h₂⟩

goal: MeasurableSet[m₁ � m₂] s

by definition: measur-
able for m1 and for m2

h₁: for m1

h₂: for m2

⟨h₁, h₂⟩

anonymous construc-
tor: builds the pair

closes

